Здравствуйте, пытаюсь вытащить из регистра записи по начислениям нескольких сотрудников и добавить их в массив. Все вроде получается, но возник вопрос, как получить только различные начисления?
К примеру в начислении зарплаты несколько сотрудников с одинаковым видом начислений "Оплата по окладу", соответственно в массив это начисление попадает несколько раз, а нужно только 1 раз (просто что оно есть).
P.S. Заранее спасибо за помощь
---
Если по факту, то можно так:

```1c
    // Удаляет повторяющиеся элементы массива.
    //
    // Параметры:
    // Массив - Массив - массив произвольных значений.
    //
    // Возвращаемое значение:
    // Массив - коллекция уникальных элементов.
    //
    Функция СвернутьМассив(Массив) Экспорт
    Результат = Новый Массив;
    ДополнитьМассив(Результат, Массив, Истина);
    Возврат Результат;
    КонецФункции

    // Дополняет массив МассивПриемник значениями из массива МассивИсточник.
    //
    // Параметры:
    // МассивПриемник - Массив - массив, в который необходимо добавить значения.
    // МассивИсточник - Массив - массив значений для заполнения.
    // ТолькоУникальныеЗначения - Булево - если истина, то в массив будут включены только уникальные значения.
    //
    Процедура ДополнитьМассив(МассивПриемник, МассивИсточник, ТолькоУникальныеЗначения = Ложь) Экспорт

    Если ТолькоУникальныеЗначения Тогда

    УникальныеЗначения = Новый Соответствие;

    Для Каждого Значение Из МассивПриемник Цикл
    УникальныеЗначения.Вставить(Значение, Истина);
    КонецЦикла;

    Для Каждого Значение Из МассивИсточник Цикл
    Если УникальныеЗначения[Значение] = Неопределено Тогда
    МассивПриемник.Добавить(Значение);
    УникальныеЗначения.Вставить(Значение, Истина);
    КонецЕсли;
    КонецЦикла;

    Иначе

    Для Каждого Значение Из МассивИсточник Цикл
    МассивПриемник.Добавить(Значение);
    КонецЦикла;

    КонецЕсли;

    КонецПроцедуры
```

Но гораздо лучше проверять при добавление элемента, есть ли он уже в массиве
---
Какие поля выбираются?
---
В запросе можно указать РАЗЛИЧНЫЕ
---
В запросе да, но мне нужно вытаскивать по каждому сотруднику отдельно, а для каждого отдельного сотрудника они и так различны
---
Если по факту, то можно так:

```1c
    // Удаляет повторяющиеся элементы массива.
    //
    // Параметры:
    // Массив - Массив - массив произвольных значений.
    //
    // Возвращаемое значение:
    // Массив - коллекция уникальных элементов.
    //
    Функция СвернутьМассив(Массив) Экспорт
    Результат = Новый Массив;
    ДополнитьМассив(Результат, Массив, Истина);
    Возврат Результат;
    КонецФункции

    // Дополняет массив МассивПриемник значениями из массива МассивИсточник.
    //
    // Параметры:
    // МассивПриемник - Массив - массив, в который необходимо добавить значения.
    // МассивИсточник - Массив - массив значений для заполнения.
    // ТолькоУникальныеЗначения - Булево - если истина, то в массив будут включены только уникальные значения.
    //
    Процедура ДополнитьМассив(МассивПриемник, МассивИсточник, ТолькоУникальныеЗначения = Ложь) Экспорт

    Если ТолькоУникальныеЗначения Тогда

    УникальныеЗначения = Новый Соответствие;

    Для Каждого Значение Из МассивПриемник Цикл
    УникальныеЗначения.Вставить(Значение, Истина);
    КонецЦикла;

    Для Каждого Значение Из МассивИсточник Цикл
    Если УникальныеЗначения[Значение] = Неопределено Тогда
    МассивПриемник.Добавить(Значение);
    УникальныеЗначения.Вставить(Значение, Истина);
    КонецЕсли;
    КонецЦикла;

    Иначе

    Для Каждого Значение Из МассивИсточник Цикл
    МассивПриемник.Добавить(Значение);
    КонецЦикла;

    КонецЕсли;

    КонецПроцедуры
```

Но гораздо лучше проверять при добавление элемента, есть ли он уже в массиве
---
Спасибо, все прекрасно работает!)
---
можно еще так попробовать
тч = Новый ТаблицаЗначений;
тч.Колонки.Добавить("Колонка");
тч.ЗагрузитьКолонку(Массив,"Колонка");
Тч.Свернуть("Колонка");
уник_массив = тч.ВыгрузитьКолонку("Колонка");
---
кто помнит типовую функцию?
---
```1c
    Если Массив.Найти(Значение) = Неопределено Тогда
    Массив.Добавить(Значение);
    КонецЕсли;
```